

Multi Agent Systems (2023-1A) - Group 4

Application of Multi-Agent System for Migrations in World Conflicts

Ukraine Refugee Crisis

University of Twente

Dion van Rijswick - s2171872

Frederico Lopes - s3096785

Pedro da Silva - s2799057

Rafael Ribeiro - s3096920

Timo Lempers - s1982176

October 31, 2022

Contents

1	Introduction	2
1.1	Motivation	2
2	Research Context	3
2.1	Ukraine	3
2.2	Related Work	4
3	Simulation	5
3.1	Initial Approach	5
3.2	Models and Diagrams	5
3.2.1	Tropos Model	5
3.2.2	Sequence Diagrams	6
4	Implementation	9
4.1	Environment	9
4.2	Refugee Agent	9
4.2.1	Lifespan	9
4.2.2	Goals	9
4.2.3	Interactions	10
4.3	Country Agent	10
4.3.1	Lifespan	11
4.3.2	Goals	11
4.4	Environment Interface	11
4.4.1	Agent World	12
4.4.2	Setup	12
4.4.3	Runtime	12
4.4.4	Monitors	13
4.4.5	Plots	13
5	Results Analysis	14
6	Conclusion	17
6.1	Future Work	17

1 Introduction

On the 20th of February 2022, war emerged in Europe with the invasion of Russia on Ukraine, and with it, millions of refugees seeking to flee the conflict and find a, new, safe place to live. Able countries have an obligation to help with the crisis, according to EU legislation [1]. However, it should also be taken into account how prepared these countries are to receive an unexpected influx of inhabitants and how their population reacts to this new situation. For this reason, we decided to do a multi-agent systems evaluation of how the different decisions of the European governments may influence the distribution of the refugees around Europe. This problem requires monitoring of the government’s decisions and tracking of the refugees, as well as measuring aspects like population discontentment as the crisis progresses. Are the European governments dealing with this refugee crisis in the best way possible?

1.1 Motivation

Human refugees have existed for ages. Whether because of war or suppression, there have always been reasons for humans to seek refuge. In the latest years, the number of involuntary migrants has increased rapidly. According to the Office of the United Nations High Commissioner for Refugees (UNHCR) [2], there were more than 100 million forcibly displaced persons until May 2022. Now with the war between Russia and Ukraine, where all information is shared through the internet, the topic is more relevant than ever before. When refugees are forced to leave their home because of a disaster, not every country or community deals with harboring them in the same way. In the Netherlands [3], for example, the population is divided into helping the refugees by taking them into their home or even driving to Ukraine to help in travels, while others do not mind the refugees coming to the Netherlands as long as they themselves are not troubled by it. Another characteristic of this problem is the cascading nature of the responsibilities that come attached: ministers make decisions on arriving refugees’ city allocations, while city mayors then need to deal with the consequences of these rulings. For example, in maintaining quality of living for both residents and aliens, or preserving local economic balance and job opportunity.

2 Research Context

2.1 Ukraine

This report analyses the migration of people from a country where armed conflicts and unsafe living conditions are persistent. Historical evidence suggests that civilians are the most vulnerable victims of war and other armed conflicts, due to their lack of sufficient military training. In several occasions, civilians have had to leave everything behind to seek refuge in safer areas.

But the Ukraine situation is far from a simple one. Ukraine possesses the biggest migration corridor in Europe [4], shared with Russia. The documentation of the migration of Ukrainian citizens over time has several gaps, most due to the lack of documentation [4] from the migrants entering Russia, whose laws allow for visa free entrance. The migrant status differs from the refugee. The Ukrainian migrant has looked for better opportunities for their family and decided other countries/places could be a defining factor to fulfill this dream, as noted in [4] [5] [6].

As of February 24th, Russia invaded Ukraine. From there on, according to UNHCR [7] over 7 million Ukrainians have sought refuge. A response plan was created, drafting the closest countries to help assist in this situation, as seen on figure 1.

Country	Data Date	Refugees from Ukraine registered for Temporary Protection or similar national protection schemes	Refugees from Ukraine recorded in country	Border crossings from Ukraine*	Border crossings to Ukraine**
Bulgaria	10/25/2022	142,806	55,257	Not applicable	Not applicable
Czech Republic	10/25/2022	452,911	453,103	Not applicable	Not applicable
Hungary	10/25/2022	31,290	31,290	1,628,968	Data not available
Poland	10/25/2022	1,469,032	1,469,032	7,113,589	5,090,535
Republic of Moldova	10/25/2022	Not applicable	95,728	670,158	313,078
Romania	10/23/2022	76,155	86,206	1,426,392	1,120,077
Slovakia	10/25/2022	98,572	98,770	883,374	620,527
Total		2,270,766	2,289,386	11,722,481	7,144,217

Figure 1: Refugee Response Plan

Nonetheless, also other countries in Europe volunteered to help and received refugees. The entrance of the Ukrainian refugees in Europe was facilitated thanks to the withdrawn of regular border controls and facilitated work visas [8]. This where these countries:

Country	Data Date	Refugees from Ukraine registered for Temporary Protection or similar national protection schemes	Refugees from Ukraine recorded in country
Albania	10/25/2022	Not applicable	2,622
Armenia	10/25/2022	Not applicable	489
Austria	10/25/2022	84,756	84,756
Azerbaijan	10/11/2022	Not applicable	3,925
Belgium	10/24/2022	57,830	58,537
Bosnia and Herzegovina	10/25/2022	Not applicable	131
Croatia	10/25/2022	18,929	18,931
Cyprus	10/25/2022	17,633	14,404
Denmark	10/25/2022	34,411	36,449
Estonia	10/25/2022	37,662	60,980
Finland	10/25/2022	42,286	38,588
France	9/26/2022	105,000	105,000
Georgia	10/11/2022	Not applicable	26,793
Germany	10/8/2022	815,134	1,002,668
Greece	10/25/2022	19,997	19,997
Iceland	10/25/2022	1,766	1,766
Ireland	10/25/2022	60,508	55,144
Italy	10/19/2022	160,673	171,546
Latvia	10/25/2022	41,812	40,163
Liechtenstein	10/25/2022	403	403
Lithuania	10/25/2022	68,925	68,925
Luxembourg	10/25/2022	6,756	6,756
Malta	9/13/2022	1,420	1,518
Montenegro	10/4/2022	6,647	27,576
Netherlands	9/30/2022	79,250	79,250
North Macedonia	10/25/2022	Not applicable	6,160
Norway	10/25/2022	30,570	30,570
Portugal	10/3/2022	52,875	52,970
Serbia and Kosovo: S/RES/1244 (1999)	10/25/2022	1,040	21,330
Slovenia	10/25/2022	7,895	8,308
Spain	10/24/2022	149,213	149,213
Sweden	10/25/2022	47,376	47,376
Switzerland	10/25/2022	66,812	66,812
Türkiye	5/19/2022	Not applicable	145,000
United Kingdom	10/18/2022	138,400	138,400
Total		2,155,979	2,593,456

Figure 2: Refugee Other Countries

Given the level of complexity of this situation, some of the Ukrainians have also fled to Russia, mostly due to separated families. The following shows the numbers of Ukrainians estimated to have entered Russia.

Country	Data Date	Refugees from Ukraine registered for Temporary Protection or similar national protection schemes	Refugees from Ukraine recorded in country	Border crossings from Ukraine*	Border crossings to Ukraine**
Belarus	10/24/2022	Not applicable	15,932	16,705	Data not available
Russian Federation	10/3/2022	Not applicable	2,852,395	2,852,395	Data not available
Total		Not applicable	2,868,327	2,869,100	Data not available

Figure 3: Refugee Russia

These figures were extracted from UNHCR's data as of October 2022.

Our goal is to simulate how the refugees will be displaced through neighbouring countries given the different policies applied to their entrance and the level of aggression encountered in their home country.

2.2 Related Work

Most of the data available on the refugee crisis and human migration is found as on UNHCR's platform. There is an interesting project called *therefugeeproject* [9], which shows the historical data on refugees worldwide.

On a closer perspective, Philip [10] has simulated the migration topic of Ukraine in the Netherlands but at an earlier period, 2015.

The closest model to our idea is seen on the proposed generalized simulation of a refugee crisis [11], which goes in depth on the refugee behaviour and prediction techniques.

3 Simulation

3.1 Initial Approach

Through agent-based modeling, we aim to emulate how the individual decisions taken influence the lives of many, finding ways to drive decision making to address these issues and enhance the quality of life of everyone involved.

There are two main groups of agents in the system, the **common people** being one, and the **country leaders** (or decision makers) being another. The common people agents will be able to communicate between themselves to deliberate over where they might be able to seek refuge based on the principles laid before them by the country's agents. In the same way that they could warn others over areas to avoid.

Historically, refugees would not have much choice in terms of where to flee to and would seek refuge in neighboring zones, but "the past decades have seen a trend towards greater geographic diffusion" [12], mostly due to better travelling conditions and access. This fact will play an important part in our simulation.

The country agents will be able to make individual choices based on what they deem important. Would they be willing to take refugees, have they made choices that would make their inhabitants seek refuge elsewhere. The effects of the refugees on the countries can be studied as well to see if they were prepared to handle it or not. These country agents don't necessarily communicate with other agents of its kind but it would be possible to come up with group solutions.

Lastly, depending on the progress, other aspects like public opinion, distribution of skilled workers inside refugee groups and supply / demand of resources could also be included.

3.2 Models and Diagrams

3.2.1 Tropos Model

The development of the model began by the definition of the actors and goals needed. The process to obtain this model was iterative, as many abstractions from the real world scenario were needed.




Figure 4: Tropos actor and goal diagram

The final diagram describes the possible actions that each agent can take.

Refugees' main goal (referenced by the quality *Individual Happiness*) is made by moving to a new country. The quality of the chosen destination is measured by the country's "resources", like *GDP* or *Distance*. This

goal is only achieved if the destination country accepts the asylum request, which in turn will depend on the current situation of the country. Namely, the population's inherent openness to harboring refugees will impact this decision. The countries are driven to balance the reward from harboring refugees with the drawback that this burden has on the national population.

3.2.2 Sequence Diagrams




Figure 5: Refugee and Country Sequence Diagram

The first sequence diagram revolves around the two main agents in the simulation: the refugees and the countries.

The refugees have to seek asylum by sending a request to a country. This request will consist of a list of information about the refugee themselves. This can include Age, Sex, Languages, Family situation, Expertise and Education level. Different combinations of these factors can be put into the asylum request based on information that the receiving country is asking for. Once the request has been sent by the refugee, the country will return a form of acknowledgement to show that the request is being processed.

Based on the request, the country has the option to refuse or accept the refugee into the country. In the denial or acceptance message to the refugee, the result is accompanied by a reason for the decision. This is important so that the refugee can perhaps make changes to their request if information is missing or other fixable problems had arisen.

This sequence diagram is most pertinent to the final simulation in question as the third agent in the sequence diagrams (Information Source) is not modeled.

Refugee Simulation: Agents Sequence Diagram

Timo Lempers | October 26, 2022




Figure 6: Refugee and News Source Sequence Diagram

This third agent, the Information source, can be found in figure 6. Although not modeled into the simulation, in a real life situation it would serve as bridge for communication between the other two agents, especially with the refugees. The information agent does not represent just a singular entity, but a combination of news articles, live news, the internet and more. The communication starts with the refugee, it will seek out the entry requirements for refugees to a certain country. The information source will then return that information. If the refugee fits those requirements, the communication continues.

The refugee will then ask how to travel to that country, and how to request asylum there. For both of the questions, the information source will reply with any information they have on the subject.

Refugee Simulation: Agents Sequence Diagram

Timo Lempers | October 26, 2022




Figure 7: Country and News Source Sequence Diagram

Lastly, the communication between the information source and countries also have to be modeled, this can be seen in figure 7. The communication between these two agents is only one directional, from the country to the information source. Any updates that a country would make regarding their policies, will be relayed to the information source, so that they may further spread it to its audience.

4 Implementation

4.1 Environment

For the development environment we chose NetLogo [13], mainly for its simplicity and approachable syntax, since it met our initial requirements. Some of the features that were used were the plotting of agent information, the interactive interface for runtime control of the simulation and the accessible agent communication language.

4.2 Refugee Agent

The mass of the agents in the simulation will be refugee persons, who leave their origin country to travel to a safe haven. They choose a destination and must go through the process of applying for asylum and travelling to the border. This process is repeated should their application be refused.




Figure 8: A female and a male refugee agents, respectively

4.2.1 Lifespan

The number of refugees that depart per tick is based on how aggressive the attacking country is (`aggression_level` variable). For every created refugee, they are randomly assigned a number to represent the minimum "aggression value" needed for them to leave the country. This value also decreases slightly over time to simulate inhabitants becoming more uncomfortable in their situation. All the agents that decided to stay will continuously review their decision.

When the decision to leave is made, the agents take into account factors such as distance, country conditions and family situation to rank their possible choices, after which they begin moving towards their chosen haven. Upon arrival they may be refused entry, at which point they must recalculate the ranking and travel to the next best choice to repeat the process. Otherwise, if accepted, the agent finishes its actions and from this point forward will only serve to influence its family member's rankings.

4.2.2 Goals

The objective of these agents is, firstly, to find a country that will accept them. Affecting the choice of which country to seek asylum in are the following factors, ordered by their weight in the objective function:

1. Number of relatives living in destination country
2. Distance from origin to destination country
3. GDP of destination country
4. Destination country's population's openness to refugees

With these factors, each refugee will rank their choices and apply to as many countries of this list as they need to until they are accepted somewhere.

One should note that this ranking is based on our understanding of war refugee's priorities, but the decision of having the family situation be the most impactful factor was taken in order for the simulation to be more dynamic and less pre-determined. Studies [14] indicate that in dangerous situations, people prioritize their economic stability over all other factors, so this priority ranking is an aspect of our work where more research and experimentation is needed to achieve a reliable model.




Figure 9: Family links examples

4.2.3 Interactions

All the interactions involving the refugees are related to the destination countries. When close enough, agents will ask to enter and the country makes a decision on whether to let them in or not.

The family relations between agents, despite not being a direct interaction, constitute an important part of their decision. At their birth, each refugee will be assigned a random number of family members (between 0 and `max_family_size`). When choosing the target country, the agent asks how many family members of his have already arrived at said country and, with this, direct his choice towards reuniting with his brood.

4.3 Country Agent

Representing both the country affected by the conflict and the neighboring countries that will aid in the refugee crisis, these agents do not move and are only in charge of deciding when refugee agents leave or enter it's borders.




Figure 10: A home and destination country agents, respectively

4.3.1 Lifespan

All instances are created at setup, with the destination countries starting with `max_refugees` values that add up to 50% of the total refugee population.

When half of the country's refugee capacity has depleted, it reviews its policy and increases this capacity based on how well the population is dealing with the flow of refugees (`population_unreceptiveness` variable). The same review happens when the capacity gets to 1/4.

Finally, when the countries get to full capacity, all other refugees attempting to enter will be refused.

The originating country starts with the entirety of the refugee population inside and as the perceived danger increases, more agents pass their tolerance threshold and depart. The only other interaction from part of this country is the closing of the borders for young male citizens, in the event of mandatory military service.

The attacking country (which isn't represented as an agent) is comprised only by the `agression_level` variable.

4.3.2 Goals

The goal of the receiving countries is to shelter refugees while maintaining their own population happy. This is done by controlling the entry of refugees based on the openness of the population to them.

4.4 Environment Interface




Figure 11: NetLogo Interface

4.4.1 Agent World

On the main screen of the environment is where the agents live for the entirety of the simulation. At the center lies the home country of the refugees and scattered around it are the destination country agents. As refugees leave their home and enter into the various neighboring countries, the labels displaying how many available spaces remain are updated.




Figure 12: Agent World screen

4.4.2 Setup

To start the simulation, destination countries must be created and the size of the refugee population must be specified, along with some other key parameters. For this, the user can either provide a file with all the values for each country or, using a number of interactable components, limit how these values are randomized. These can be edited before the simulation is setup, as follows.

In figure 13, the horizontal sliders control the characteristics of both country and refugee agents. The vertical sliders below control the weight of each factor in the refugee agent's decision function. Finally, the button below creates the agents with the given parameters so the simulation can be started.




Figure 13: Setup buttons and sliders

4.4.3 Runtime

These controls are meant to be interacted with while the simulation is in progress, for various reasons. Both the `agression_level` slider and the `mandatory_military` switch control aspects of the war scenario, like

the names indicate. The bottom button toggles if the family links between refugees are visible or not, and the go button starts and pauses the simulation.




Figure 14: Runtime buttons and sliders

4.4.4 Monitors

Used to display the properties of destination countries. Since they provide an easy way to inspect these values at setup and throughout the simulation, they were useful during section 5.




Figure 15: Country variables monitors

4.4.5 Plots

Plots showing the evolution of the situation were used as a means to draw conclusions. The plots with the colors legend on the side display said values over time for each destination country. These are further analysed in section 5.




Figure 16: Various plot graphs

5 Results Analysis




Figure 17: Graph of departed refugees with no mandatory military participation




Figure 18: Graph of departed refugees with mandatory military participation being lifted

In figure 17 and 18, all parameters are kept at a constant, except for the fact that in figure 17, no mandatory military participation is enacted throughout the entirety of the run. In figure 18 we see that, from the start of the simulation until approximately half way, mandatory military participation is enacted. Shortly after refugees stop leaving the home country, the mandatory military rule is stopped. This can be seen by looking at the second curve upwards again after around 400 ticks.

Although the same amount of refugees leave the country in total, the time taken to all depart the country is much longer due to having two waves of refugees leave at different times. Due to the majority of refugees having left before the mandatory military was lifted, the second wave was also at a slower rate of leaving.

In a real life situation, the level of aggression from other countries is not a stable value, it can fluctuate. Towards the end of a conflict, the danger in a country might already be going down, and if a conflict just starts, the level of danger can still be considered very low. This change in aggression level can be simulated. Aggression level in the country is quantified by a value ranging from 0-100. This value can be changed throughout the course of the simulation running.

In figure 19, this change can be seen. The level of aggression starts out at a low value of just 10, but over the course of the simulation, the level of aggression reaches its highest, 100. A curve can be seen becoming steeper and steeper at a similar rate as the aggression level going up. This is visible in all 4 of the different graphs showcasing how they relate to one another.

Slowly Increasing Aggression




Figure 19: Graphs showcasing a slowly increasing rate of aggression

There are also four parameters detailing the refugees main objectives: the four vertically sliders seen in 13. These parameters are set before the simulation is started and have an effect on which country the refugees will attempt to seek asylum in.

For example, if the refugees take a max importance in the GDP of a country and extremely low importance in the other parameters, a simulation like 20 can take place. In this case, almost every single refugee is going to the exact same country.

To contrast this example with another parameter. The same simulation can be run, but instead of a max importance on GDP of a country, you take the Max Openness of the country instead and leave all the other parameters the same. Suddenly the simulation and graph in 21 look very different as the refugees are all choosing a completely different country instead.




Figure 20: Simulation with the GDP parameter at its highest




Figure 21: Simulation with high openness importance

However, these aren't realistic simulations at all. In real life situations, a single factor might be the most important for people, but it's not just a single factor that is deciding the entire choice. A mix of the different parameters is what will end up happening. For the next simulation, a mix of different parameters was used, as can be seen in 22. In the previous two simulations, all refugees chose a singular country to move too. However in this new simulation, there is a much larger spread of initial countries that is being chosen. This is because some countries might specialize in a certain aspect and since the refugees value different aspects, different countries are chosen.

The corresponding graph also shows the same information. A spread of countries being moved to directly from the start instead of only towards the end of the simulation.




Figure 22: Simulation with mixed parameters

6 Conclusion

Unfortunately, humanitarian crises that create a big flow of refugees are still a very real problem, and because of that we hope this and some more future works that might follow, can help governments, organizations, and people to recognize the scale of the problem and what should be done in a situation like this.

In conclusion, our work and simulation serves as a way of estimating how a humanitarian tragedy can affect the country and the countries around. We can analyze the flow of the refugees and the impact that the government decisions have on them, having this way a basis of what should actually be done in a real-life.

6.1 Future Work

In a next iteration we could add things like children school systems since countries can only accept refugee children if they have enough space in schools for them. Another interesting addition would be a way of controlling each countries' policies separately so that we could better study how a country decision influences the countries around and the flow of refugees. This was not possible mostly due to limitations with the simulation platform. Some aspects of the simulation can also be improved. For example the family relationships don't always impact the refugees decisions to the extent that is wanted. Finally, there should be a way of easily importing real world data for the setup of the simulation, like reproducing the world map and countries' parameters. Currently, controlling the initial setup through a text file is not very practical.

References

- [1] “Common european asylum system.” https://home-affairs.ec.europa.eu/policies/migration-and-asylum/common-european-asylum-system_en.
- [2] “Global displacement hits record capping decade long rising trend.” <https://www.unhcr.org/news/press/2022/6/62a9d2b04/unhcr-global-displacement-hits-record-capping-decade-long-rising-trend.html>.
- [3] “Safety regions municipalities and private individuals offer refugees a safe place to stay.” <https://www.government.nl/topics/reception-of-refugees-from-ukraine/news/2022/04/07/safety-regions-municipalities-and-private-individuals-offer-refugees-a-safe-place-to-stay>.
- [4] Y. Leontiyeva, “Ukrainian migration to europe: Policies, practices and perspectives,” *Central and Eastern European Migration Review*, vol. 3, pp. 5–10, 2014.
- [5] F. A. Vianello, “Ukrainian migrant workers in italy: Coping with and reacting to downward mobility,” *Central and Eastern European Migration Review*, vol. 3, pp. 85–98, 2014.
- [6] V. Marchese, B. Formenti, N. Cocco, G. Russo, J. Testa, F. Castelli, and M. Mazzetti, “Examining the pre-war health burden of ukraine for prioritisation by european countries receiving ukrainian refugees,” *The Lancet Regional Health - Europe*, vol. 15, p. 100369, 2022.
- [7] “Ukraine.” https://data.unhcr.org/en/situations/ukraine#_ga=2.21007931.1425181503.1667142917-1278501816.1666775368.
- [8] A. D. Näre, L. and N. Irastorza, “What can we learn from the reception of ukrainian refugees?,” 2022.
- [9] “The refugee project.” <https://www.therefugeeproject.org/>.
- [10] P. Wozny, “A value sensitive agent based simulation of the refugee crisis in the netherlands,” 2018.
- [11] B. D. . G. D. Suleimenova, D., “A generalized simulation development approach for predicting refugee destinations..” <https://doi.org/10.1038/s41598-017-13828-9>.
- [12] “The globalization of refugee flows.” <https://documents1.worldbank.org/curated/en/184471586279885821/pdf/The-Globalization-of-Refugee-Flows.pdf>.
- [13] “Netlogo, a multi-agent programmable modeling environment.” <https://ccl.northwestern.edu/netlogo/>.
- [14] P. Adhikari, “Conflict-induced displacement, understanding the causes of flight.” <https://onlinelibrary.wiley.com/doi/pdf/10.1111/j.1540-5907.2012.00598.x>.

Appendices

Appendix A- Netlogo Code

```
globals [
  show_family_links?
  transparent
  max_age
  max_refugee_number
  total_refugees_departed
  gender_list
  first_refugee_wave

  max_leaving_delay
  max_population
  min_population

  gdp_min
  gdp_max
  pop_un_min
  pop_un_max
]

;; agents
breed [refugees refugee]
breed [countries country]

;; refugee agent links
undirected-link-breed [family-links family-link]
family-links-own [separated?]

;; link between refugee and country with number of family members that arrived there
undirected-link-breed [family-at-links family-at-link]
family-at-links-own [n_family]

countries-own[
  first_update
  second_update
  gdp ; divided by 1k
  population_unreceptiveness
  population
  max_refugees
  is_starting_country?

  accepted_number
  refused_number
  reunions_number
]

refugees-own[
  age
  gender
  race
  target_country
  visited_countries
  likeliness_of_staying
  min_likeliness_of_staying
  family_size
  family_remaining

  moving?
  departed?
  arrived?
  leaving_delay

  openness_weight
  distance_weight
  family_distance_weight
  gdp_weight
]

to setup
  clear-all
  reset-ticks
  file-close-all

  setup-refugees
  setup-countries
  new-refugees
  setup-interface
  update-label

  reset-ticks
end

to setup-interface
  ; reset runtime interface
  set aggression_level 10
  set mandatory_military false

  set show_family_links? false
  Toggle-show-links

  set transparent [0 0 0 0]
end

to setup-refugees
  set max_age 100
  set max_refugee_number refugee_population
  set total_refugees_departed 0
  set gender_list ["Male" "Female"]
```

```

end

to setup-countries
  set max_leaving_delay 1000
  set max_population 3000
  set min_population 200

  set gdp_min 500
  set gdp_max 3000
  set pop_un_min 30
  set pop_un_max 70

  let locations_list []
  let gdp_list []
  let population_unreceptiveness_list []
  let color_list []
  let population_list []

  ;; load input scenario from file
  let random_params? input_file = ""
  if not random_params? [
    file-open (word "inputs/" input_file)
    set locations_list file-read
    set gdp_list file-read
    set population_list file-read
    set population_unreceptiveness_list file-read
  ]

  let j 0

  let patches_list []
  ifelse random_params?[
    let patches_agentset n-of number_receiving_countries patches with [
      distancexy 0 0 > 4 and abs pxcor < (max-pxcor - 1) and
      abs pycor < (max-pycor - 1)
    ]
    set patches_list [self] of patches_agentset
  ][
    set patches_list map [
      coordinates -> patches with [
        pxcor = item 0 coordinates and
        pycor = item 1 coordinates
      ]
    ] locations_list
  ]
  foreach patches_list [
    this_patch -> (
      ask this_patch [
        ; in order to get patches by input file order

        set color_list lput ((3 + random-float 6) + (10 * random 14)) color_list
        if random_params? [
          set population_list lput (min_population + (random (max_population - min_population))) population_list
        ]

        sprout-countries 1 [
          set shape "box"
          set size 2
        ]
        set j j + 1
      ]
    )
  ]
  let total_population sum population_list

  let i 0
  repeat count countries [
    ask country i [
      set label max_refugees
      set color item i color_list

      set population item i population_list
      set max_refugees round (population / total_population * refugee_population * 0.9)

      ifelse random_params? [
        set gdp gdp_min + (random (gdp_max - gdp_min))
        set population_unreceptiveness pop_un_min + (random (pop_un_max - pop_un_min))
      ][
        set gdp item i gdp_list
        set population_unreceptiveness item i population_unreceptiveness_list
      ]

      set first_update false
      set second_update false
      set accepted_number 0
      set refused_number 0
      set is_starting_country? false
    ]
    set i i + 1
  ]
  create-countries 1 [
    set shape "house"
    set size 2
    set label population
    set color ((3 + random-float 6) + (10 * random 14))

    set population refugee_population
    set is_starting_country? true
  ]
end

;; called every tick
to go
  accept-refugee
  move-refugees
  review-refugees
  review-policies

```

```

update-label
tick
end

;; "creates" the refugees
to new-refugees
create-refugees max_refugee_number [
  ; properties
  set moving? false
  set arrived? false
  set departed? false
  set likeliness_of_staying random 100
  set min_likeliness_of_staying random likeliness_of_staying
  set age random max_age
  set gender one-of gender_list
  set target_country nobody
  set visited_countries []
  set family_size max list ((random (max_family_size * 2)) - max_family_size) 0
  set family_remaining family_size
  set leaving_delay random max_leaving_delay
  set openness_weight 1 + random (max_openness_weight - 1)
  set distance_weight 1 + random (max_distance_weight - 1)
  set family_distance_weight 1 + random (max_family_distance_weight - 1)
  set gdp_weight 1 + random max_gdp_weight
  ; visuals
  set shape "person"
  ifelse gender = "Male"[
    set color 105
  ][
    set color 135
  ]
  if age < 18[
    set size 0.7
  ]
]

; creating family links between refugees
ask refugees[
  ; select n random refugees, create links and update their remaining family_size
  let possible_family_members other refugees with [family_remaining > 0]

  ; if no more people left with family to link, return
  if count possible_family_members <= 0 [stop]

  set family_remaining 0
  let random_refugees possible_family_members
  if family_size <= count possible_family_members [
    set random_refugees (n-of family_size possible_family_members)
  ]

  create-family-links-with random_refugees [
    set color transparent
  ]
  ask random_refugees [
    set family_remaining family_remaining - 1
  ]
]
end

to review_refugees
ask refugees[
  ; if refugee hasn't left yet
  if not arrived? and not moving? [
    if likeliness_of_staying > min_likeliness_of_staying[
      set likeliness_of_staying likeliness_of_staying - random-float 0.005
    ]
    ifelse likeliness_of_staying < aggression_level[
      ; mandatory enrollment prevents young males from leaving
      ifelse ( mandatory_military = true) and (gender = "Male") and (age > 18) and (age < 65) [
        set moving? false
      ][
        set moving? true
      ]
    ][
      set moving? false
    ]
  ]
]
end

;; moves the refugees
to move-refugees
ask refugees with [moving?] [
  ifelse leaving_delay > 0[
    set leaving_delay leaving_delay - 1
  ][
    ifelse target_country = nobody[
      choose_country
    ][
      let move_distance random-float 1
      face target_country
      forward move_distance

      if not departed? and move_distance > 0[
        set departed? true
        set total_refugees_departed total_refugees_departed + 1
      ]
    ]
  ]
]
end

;; reviews country policies based on parameters
to review-policies
ask countries [

```

```

if not is_starting_country? [
  if (accepted_number / max_refugees) > 0.5 [
    if (first_update != true) [
      set max_refugees round(max_refugees * (1 / (population_unreceptiveness / 100)))
      set first_update true
    ]
  ]
  if (accepted_number / max_refugees) > 0.75 [
    if (second_update != true) [
      set max_refugees round(max_refugees * (1 / (population_unreceptiveness / 100)))
      set second_update true
    ]
  ]
]
]
end

to choose_country
ask refugees with [moving? and target_country = nobody] [
  let refugees_visited_countries visited_countries
  let family my-family-links
  let temp_openness_weight openness_weight
  let temp_distance_weight distance_weight
  let temp_gdp_weight gdp_weight
  let temp_family_distance_weight family_distance_weight
  let temp_family_size family_size
  let desired_list []
  let country_list []
  ask countries[
    let this_country self

    ifelse
    ; refugees can only move to a country other than their home
    not is_starting_country?
    ; refugees won't reattempt to enter a visited country
    and not (member? self refugees_visited_countries)
    ; if country is full nobody will choose it anymore
    and max_refugees - accepted_number > 0[

      let openness_weighted temp_openness_weight * ((population_unreceptiveness - 70) / 30) * ((population_unreceptiveness - 70) / 30)

      let distance_to_refugee distance self
      let distance_weighted temp_distance_weight * (distance_to_refugee / 40) * (distance_to_refugee / 40)

      ; if a family-at-link exists between this country and refugee
      let family_count 0
      if in-family-at-link-neighbor? myself
      [
        ; family count is n_family of said link
        set family_count [n_family] of (family-at-link (who) ([who] of myself))
      ]

      ;if family_count > 0 [show [who] of myself show family_count]

      let family_distance_weighted 0
      if temp_family_size > 0 [
        set family_distance_weighted temp_family_distance_weight * ((temp_family_size - family_count) / temp_family_size) * ((temp_family_size - family_count) / temp_family_size)
      ]

      let gdp_weighted temp_gdp_weight * ((3000 - gdp) / 2500) * ((3000 - gdp) / 2500)

      let value sqrt(distance_weighted + openness_weighted + family_distance_weighted + gdp_weighted)
      set desired_list lput value desired_list
      set country_list lput this_country country_list
    ]
    [
      ; baseline desirability
      set desired_list lput 10000 desired_list
      set country_list lput this_country country_list
    ]
  ]

  ; target country will be the one with lowest value in desired_list
  let desired_index min desired_list
  let desired_country position desired_index desired_list
  set target_country (item desired_country country_list)

  ; update family links
  if target_country != nobody [
    ask my-family-links [
      ; if both ends of the link have different target countries, they are separated
      if [target_country] of other-end != [target_country] of myself [
        set separated? true
        set color red
        set thickness 0.1
      ]
    ]
  ]
]
end

to accept-refugee
let accepted? false
ask refugees with [moving? and not (target_country = nobody)] [

  if distance target_country < 0.25[
    ;check accepted
    set visited_countries lput target_country visited_countries
    ask target_country[
      ifelse accepted_number < max_refugees[
        set accepted? true
        ;;set moving? false
        set accepted_number accepted_number + 1
      ]

      ; save number of refugees that have reunited with family members

```

```

        let link_to_target_country family-at-link (who) ([who] of myself)
        if not (link_to_target_country = nobody) and [n_family] of link_to_target_country > 0[
            set reunions_number reunions_number + [n_family] of link_to_target_country
        ]
    ]
    [set refused_number refused_number + 1]
]
if not accepted? [
    set target_country nobody
    stop
]

set moving? false
set arrived? true
set accepted? false

; update family links with reunions where needed
ask my-family-links [
    let arrival_country_link [target_country] of myself

    ; Both family member and this refugee have same target countries
    if [target_country] of other-end = arrival_country_link[
        ifelse [moving?] of other-end[
            ; if one is moving, they are reuniting
            set color green
            set thickness 0.5
        ][
            ; if they have both arrived, they are reunited
            if [arrived?] of other-end [
                set separated? false
                set color transparent
            ]
        ]
    ]
]

; for each family member,
; update/create family-at-links as needed
ask other-end [
    ; if family-at-link exists between family member -- target country
    ifelse in-family-at-link-neighbor? arrival_country_link
    [
        ; increment n_family
        ask family-at-link (who) ([who] of arrival_country_link) [
            set n_family n_family + 1
        ]
    ]
    [
        ; otherwise, create link and set n_family
        create-family-at-link-with arrival_country_link [
            set n_family 1
            set color transparent ;these links are never visible
        ]
    ]
]
]
]
end

to update-label
ask countries [
    ifelse is_starting_country? [
        ifelse ticks > 1[
            set label round max_refugee_number - total_refugees_departed
        ][
            set label round max_refugee_number
        ]
    ][
        set label round max_refugees - accepted_number
    ]
]
end

to toggle-show-family-links
ask family-links [
    set hidden? not show_family_links?
]
end

to Toggle-show-links
toggle-show-family-links
set show_family_links? (not show_family_links?)
end

```
